PharmaSUG 2019 - Paper PO-277

# Flagging On-Treatment Events in a Study with Multiple Treatment Periods

David Franklin, IQVIA

## ABSTRACT

Typically, for data like Adverse Events, the common practice is to flag events that occur on treatment with a definition of something like "Event starting on or after date of first treatment and 30 days after date of last treatment". This fairly easy to program, but when things like a cycle of treatment may have multiple periods due to patient events, this usual approach does not work. This paper introduces the ADaM variable APHASE and how this solved a structural problem with the data, presents a macro that was very useful in producing the solution.

## INTRODUCTION

Typically, for data like Adverse Events (AE), the common practice is to flag events that occur on treatment with a definition of something like:

> Event starting on or after date of first treatment and 30 days after date of last treatment.

Using this practice it is fairly easy, programming is something like

> TRTSDT<=ASTDT<=(TRTEDT + x days) then TRTEMFL='Y';

where

> TRTSDT Treatment Start Date
>
> ASTDT Analysis Start Date (usually onset date)
>
> TRTEDT Treatment End Date
>
> TRTEMFL Treatment Emergent Analysis Flag
>
> x defined by the sponsor and often incorporates the known half-life of the treatment

If there are partial dates for a AE start date a rule similar to:

> If day is missing and month and year are the same as first treatment, set imputed start date to date of first treatment, otherwise first of the month; if day and month are missing but year is the same as first treatment, set date to date of first treatment, otherwise set to first of the year

is usually present in the Statistical Analysis Plan.

## WHAT ABOUT IF THERE ARE DIFFERENT PERIODS

But what if you have different periods for a treatment within a cycle, for example a subject may take a holiday from the treatment within the same cycle, sometimes for days or months at a time? Or the case where it is a study with different cycles of treatment? The usual practice above does not usually apply; what is present instead is

```
|<-- Period 1 -->| |<-- Period 2 -->| ...
```

where there are two different start periods in play. Usually the Statistical Analysis Plan will have something like

```
"… excludes time where treatment is not taken"
```

or similar.

How this is set up depends a lot on the study – in one study there was both 'holidays' within a cycle (line of therapy) and cycles (line of therapy); in another study, and something that is more common, different

cycles (line of therapy) separated by holidays. Whatever the structure, the solution is similar. In the second case, different cycles (line of therapy) separated by holidays, there should be variables TRxxSDT and TRTxxEDT defined in the ADSL and carried forward to the programming – these two variables define the start and stop dates of each period with xx as an incremental counter for each period.

To set a treatment emergent flag based on this approach needs the use of SAS arrays. In the example below we have a maximum of three possible treatment periods for a subject (some may take less!):

```
array zaz {*} TR01SDT TR02SDT TR03SDT;
array zbz {*} TR01EDT TR02EDT TR03EDT;
do i=1 to dim(zaz);
  *Go through each treatment period, but remember
   not everyone may have that period of treatment!
   Also if event is already treatment emergent, no
   need to check further.;
  if zaz{i)>. and TRTEMFL^='Y' then do;
    *Process partial dates. First complete date;
    if lengthn(AESTDTC)=10 then
     ASTDT=input(AESTDTC,yymmdd10.);
    *Second, missing day. If month and year match
     treatment date, set to that date, else first
     of the month;
    else if lengthn(AESTDTC)=7 then do;
      if put(zaz(i),yymmd7.)=AESTDTC then
       ASTDT=zaz(i);
      else
       ASTDT=input(cats(AESTDTC,'-01'),yymmdd10.);
      ASTDTF='D';
    end;
    *Third, missing day and month. If year is same as
     the treatment start date, set to that date, else
     set to first of the year;
    else if lengthn(AESTDTC)=4 then do; *Year Value;
      if put(zaz(i),year.)=AESTDTC then ASTDT=zaz(i);
      else ASTDT=input(cats(AESTDTC,'-01-01'),yymmdd10.);
      ASTDTF='M';
    end;
    *Finally, and you will not always see this in a
     study, if date is missing, set to treatment
     start date. This is sometimes done to assume
     that and event is treatment emergent if no date;
    else if lengthn(AESTDTC)=0 then do;
      ASTDT=zaz(i);
      ASTDTF='Y';
    end;
    *Check for date to see if emergent, if so set
     flag and signal period event occurred in;
    if .<zaz(i)<=ASTDT<=zbz(i) then do;
      TRTEMFL='Y';
      APERIOD=i;
    end
  end;
end;
```

Through this loop, if the event is not emergent then TRTEMFL and APERIOD will be blank but an imputed date will be present.

## LETS ADD PHASES

What about the case where we have numerous holidays in a cycle with subjects taking many cycles, e.g.

```
|<-- Phase 1 -->|  |<--           Phase 2         -->  | ...
|<-- Period 1 -->| |<-- Period 2 -->| |<-- Period 3 -->| ...
```

In the ADaM world there is a variable called APHASE that can solve this – this can be thought of as a higher level of groups of APERIOD. As in the timeline above, a Phase can have different periods. In the case of one study, allowance was made for up to 12 drug holidays! So how would we vary the code to handle this situation? Lets look again with the example given earlier, but APHASE=1 when TR01SDT, TR01EDT, TR02SDT and TR02EDT; and APPHASE=2 when TR03SDT and TR03EDT are in play (changes are in red, below):

```
array zaz {*} TR01SDT TR02SDT TR03SDT;
array zbz {*} TR01EDT TR02EDT TR03EDT;
array zcz {3} _temporary_ (1 1 2);
do i=1 to dim(zaz);
  *Go through each treatment period, but remember
   not everyone may have that period of treatment!
   Also if event is already treatment emergent, no
   need to check further.;
  if zaz{i)>. and TRTEMFL^='Y' then do;
    *Process partial dates. First complete date;
    if lengthn(AESTDTC)=10 then
     ASTDT=input(AESTDTC,yymmdd10.);
    *Second, missing day. If month and year match treatment
     date, set to that date, else first of the month;
    else if lengthn(AESTDTC)=7 then do;
      if put(zaz(i),yymmd7.)=AESTDTC then ASTDT=zaz(i);
      else ASTDT=input(cats(AESTDTC,'-01'),yymmdd10.);
      ASTDTF='D';
    end;
    *Third, missing day and month. If year is same as
     the treatment start date, set to that date, else
     set to first of the year;
    else if lengthn(AESTDTC)=4 then do; *Year Value;
      if put(zaz(i),year.)=AESTDTC then ASTDT=zaz(i);
      else ASTDT=input(cats(AESTDTC,'-01-01'),yymmdd10.);
      ASTDTF='M';
    end;
    *Finally, and you will not always see this in a
     study, if date is missing, set to treatment
     start date. This is sometimes done to assume
     that and event is treatment emergent if no date;
    else if lengthn(AESTDTC)=0 then do;
      ASTDT=zaz(i);
      ASTDTF='Y';
    end;
    *Check for date to see if emergent, if so set
     flag and signal period event occurred in;
    if .<zaz(i)<=ASTDT<=zbz(i) then do;
      TRTEMFL='Y';
      APERIOD=i;
      APHASE=zcz(i);
    end
  end;
end;
```

All we have done here is add two lines, one being a temporary array with the APHASE values associated with each APERIOD.

Note that in this type of study construction subjects may have missing TRxxSDT/TRxxEDT combinations due to not everyone having consecutive treatment period, e.g. using the structure in the example, one subject may have TR01SDT/TR01EDT, TR02SDT/TR02EDT and TR03SDT/TR03EDT, but another subject may just have TR01SDT/TR01EDT and TR03SDT/TR03EDT combinations because they did not have a TR02SDT/TR02EDT period treatment in Phase 1.

## CONCLUSION

Typically, for data like Adverse Events, the common practice is to flag events that occur on treatment with a definition of something like "Event starting on or after date of first treatment and 30 days after date of last treatment". This paper has gone beyond this to present the concept of "Periods" and "Phases", and presented a macro that can flag on treatment events with these situations.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

David Franklin

david.franklin1@iqvia.com